

DEX ... PARETO PLOT

PURPOSE

Generates a dex Pareto plot for a given statistic.

DESCRIPTION

A dex statistic Pareto plot is a graphical method for representing a design of experiment problem. The first variable is a response variable while the remaining variables (must be at least one) represent levels of factors. The user specified statistic is computed for each level of each factor. These statistics are then ordered from high to low (this is where the Pareto term comes from). The plot consists of:

- Vertical axis = values of the computed statistic for each level of each factor ordered from high to low;
- Horizontal axis = an index value (1 for the highest value of the statistic, N for the lowest value of the statistic where N is the total number of levels in all factors).

This plot graphically shows the magnitude and spread of the given statistic for the various levels of factors.

SYNTAX 1

DEX <stat> PARETO PLOT <y> <x1> ... <xn> <SUBSET/EXCEPT/FOR qualification>

where <y> is the response variable;

<x1> ... <xn> are a sequence of variables representing factors in a designed experiment;

<stat> is one of the following statistics:

MEAN (or AVERAGE), MIDMEAN, MEDIAN, TRIMMED MEAN, WINDSORIZED MEAN,
 SUM, PRODUCT, SIZE (or NUMBER or COUNT), MINIMUM, MAXIMUM,
 STANDARD DEVIATION (or SD), VARIANCE,
 STANDARD DEVIATION OF MEAN (or SDM), VARIANCE OF MEAN (or VM),
 RELATIVE STANDARD DEVIATION (or RELSD),
 RELATIVE VARIANCE (or RELV or COEFFICIENT OF VARIATION),
 RANGE, MIDRANGE, LOWER HINGE, UPPER HINGE, LOWER QUARTILE, UPPER QUARTILE,
 <FIRST/SECOND/THIRD/FOURTH/FIFTH/SIXTH/SEVENTH/EIGHTH/NINTH> DECILE (or 1DEC, 2DEC,
 3DEC, 4DEC, 5DEC, 6DEC, 7DEC, 8DEC, 9DEC),
 SKEWNESS, KURTOSIS, PROPORTION,
 AUTOCORRELATION, AUTOCOVARANCE,
 SINE FREQUENCY, SINE AMPLITUDE,
 CP, CPK, EXPECTED LOSS, PERCENT DEFECTIVE,
 SINE FREQUENCY, SINE AMPLITUDE,
 TAGUCHI SN, TAGUCHI SN+, TAGUCHI SN-, TAGUCHI SN00;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used for statistics that only require a single variable to compute.

SYNTAX 2

DEX <stat> PARETO PLOT <y> <x> <x1> ... <xn> <SUBSET/EXCEPT/FOR qualification>

where <y> is the response variable;

<x> is a second variable used in calculating the statistic (e.g., a linear fit is computed between <y1> and <x>);

<x1> ... <xn> are a sequence of variables representing factors in a designed experiment;

<stat> is one of the following statistics:

LINEAR INTERCEPT, LINEAR SLOPE, LINEAR RESSD, LINEAR CORRELATION,

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used for statistics that require two variables to compute.

EXAMPLES

```
DEX MEAN PARETO PLOT Y X1 X2
DEX SD PARETO PLOT Y X1 X2 X3
DEX RANGE PARETO PLOT Y X1 X2 X3 X4
DEX RANGE PARETO PLOT Y X1 TO X4
```

NOTE 1

This plot is similar to the DEX ... PLOT command. The distinction is that the Pareto version sorts by the value of the statistic while the DEX .. PLOT command does not. The Pareto version loses the connection between a given value and the factor to which it belongs, so

it is typical to do a DEX ... PLOT command before a DEX ... PARETO PLOT command. The Pareto version can sometimes show more distinctly the difference in magnitude and the amount of variation in the calculated statistic.

NOTE 2

This plot is normally done for a location parameter (typically the mean or median) or a spread parameter (typically the standard deviation or range). The other statistics are less often used.

NOTE 3

The TO syntax is allowed for the list of factor variables (see the EXAMPLES above).

NOTE 4

The CHARACTER, BAR, SPIKE, and LINE settings can be used to control the appearance of the plot. Trace 1 contains the calculated value of the statistic and trace 2 contains the value of the statistic for all values of the response variable. Pareto charts are typically drawn as bars. However, you can draw them as spikes, individual points, or a connected line if you prefer. The program example below demonstrates drawing them as bars. The various plot control commands for characters, lines, spikes, and bars can be used as usual.

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

LINES	=	Sets the type for plot lines.
CHARACTER	=	Sets the type for plot characters
PLOT	=	Generates a data or function plot.
DEX SCATTER PLOT	=	Generates a dex scatter plot.
DEX SIGN PLOT	=	Generates a dex sign plot.
DEX ... PLOT	=	Generates a dex plot for a statistic.
DEX ... YOUNDEN PLOT	=	Generates a Youden dex plot for a statistic.
DEX ... EFFECTS PLOT	=	Generates a dex effects plot for a statistic.
DEX ... PARETO EFFECTS PLOT	=	Generates a Pareto effects dex plot for a statistic.
DEX ... ABSOLUTE EFFECTS PLOT	=	Generates an absolute effects dex plot for a statistic.
DEX ... PARE ABSO EFFECTS PLOT	=	Generates a Pareto absolute effects dex plot for a statistic.
DEX WIDTH	=	Specifies the width of levels in a dex plot.

REFERENCE

"Statistics for Experimenters," Box, Hunter, and Hunter, John Wiley and Sons, 1978.

APPLICATIONS

Design of Experiments

IMPLEMENTATION DATE

89/12

PROGRAM

```

SKIP 25
READ SHEESLE2.DAT Y PROC PLANT SPEED SHIFT PROC
.
BAR ON
BAR WIDTH 0.5
LINES BLANK SOLID
TITLE DEX PARETO PLOT
.
MULTIPLY 2 2
MULTIPLY CORNER COORDINATES 0 0 100 100
YILABEL MEAN
LET A = MEAN Y
BAR BASE A
DEX MEAN PARETO PLOT Y PROC PLANT SPEED SHIFT
YILABEL MEDIAN
LET A = MEDIAN Y
BAR BASE A
DEX MEDIAN PARETO PLOT Y PROC PLANT SPEED SHIFT
YLIMITS
YILABEL STANDARD DEVIATION
LET A = STANDARD DEVIATION Y
BAR BASE A
DEX SD PARETO PLOT Y PROC PLANT SPEED SHIFT
YILABEL RANGE
LET A = RANGE Y
BAR BASE A
DEX RANGE PARETO PLOT Y PROC PLANT SPEED SHIFT
END OF MULTIPLY
    
```

